
    Getting started

The package can be installed by adding confispex to your list of dependencies in mix.exs:
def deps do
  [
    {:confispex, "~> 1.2"}
  ]
end
Let's define the first version of a variables schema and runtime config with confispex:
defmodule MyApp.RuntimeConfigSchema do
  import Confispex.Schema
  @behaviour Confispex.Schema
  alias Confispex.Type

  defvariables(%{
    "LOG_LEVEL" => %{
      cast:
        {Type.Enum,
         values: [
           "emergency",
           "alert",
           "critical",
           "error",
           "warning",
           "notice",
           "info",
           "debug",
           "none"
         ]},
      groups: [:base]
    },
    "DATABASE_URL" => %{
      aliases: ["DB_URL"],
      doc: "Full DB URL",
      cast: Type.URL,
      context: [env: [:prod]],
      groups: [:primary_db],
      required: [:primary_db]
    },
    "DATABASE_POOL_SIZE" => %{
      aliases: ["DB_POOL_SIZE", "POOL_SIZE"],
      cast: {Type.Integer, scope: :positive},
      default: "10",
      context: [env: [:prod]],
      groups: [:primary_db]
    },
    "CONTACT_US_EMAILS" => %{
      cast: {Type.CSV, of: Type.Email},
      default: "help@example.com,feedback@example.com",
      groups: [:landing_page],
      context: [env: [:dev, :prod]]
    }
  })
end
You can read about all possible options in doc about the type Confispex.Schema.variable_spec/0.
Understanding Key Concepts
Before we configure runtime, let's understand some important concepts used in the schema above.
Aliases
Variables can have multiple names. Confispex tries them in order:
"DATABASE_URL" => %{
  aliases: ["DB_URL"],  # If DATABASE_URL not found, tries DB_URL
  # ...
}
This means you can set either DATABASE_URL=postgres://... or DB_URL=postgres://... and Confispex will find it.
Context Filtering
Variables can be limited to specific environments:
"DATABASE_URL" => %{
  context: [env: [:prod]],  # Only available in production
  # ...
}
When you run in :dev or :test, this variable won't appear in the schema at all. This prevents confusion about which variables apply to which environment.
Error Handling
When Confispex.get/1 encounters an error (type casting fails or variable not found), it returns nil instead of raising an exception. All errors are collected and can be viewed later:
mix confispex.report --mode=detailed

This design allows you to see ALL configuration problems at once, rather than fixing them one-at-a-time.
The Store
The "store" is where Confispex reads configuration values from. By default, it uses System.get_env/0, which returns all environment variables as a map. You can provide a custom store to read configuration from other sources like JSON files, databases, or any other data structure:
# Read from JSON file
store = File.read!("config.json") |> JSON.decode!()

Confispex.init(%{
  schema: MyApp.ConfigSchema,
  context: %{env: :prod},
  store: store
})

# Or use a function that lazily loads the store
Confispex.init(%{
  schema: MyApp.ConfigSchema,
  context: %{env: :prod},
  store: fn -> File.read!("config.json") |> JSON.decode!() end
})
Runtime Configuration
Put the following content to config/runtime.exs:
import Config

Confispex.init(%{
  schema: MyApp.RuntimeConfigSchema,
  context: %{env: config_env(), target: config_target()}
})

# application config

config :logger,
  level: String.to_atom(Confispex.get("LOG_LEVEL"))

config :my_app,
  contact_us_emails: Confispex.get("CONTACT_US_EMAILS"),
  database_pool: Confispex.get("DATABASE_POOL_SIZE"),
  database_ssl: !Confispex.get("DATABASE_NO_SSL")
Now, if you run
LOG_LEVEL=info CONTACT_US_EMAILS=myemail1@example.com,myemail2 MIX_ENV=prod mix confispex.report --mode=detailed
you'll see the following report
[image: state 1]
Group colors
	green - group has required variables and they are present and valid. Such color is not present on a screenshot above, we'll make green group later.
	red - group has required variables, they aren't present or they are invalid.
	blue - group doesn't have required variables and always functional, because there is always a default value to which system can fall back.

There are 3 groups in our example :landing_page, :base and :primary_db:
	:primary_db is not functional, because all required variables weren't provided.
	:base is functional, everything is valid.
	:landing_page is functional too, because even if system failed to cast some value, default value is present and it is used.

Symbols
	* - variable is required in specified group.
	? - variable is defined in schema, but was not invoked in runtime.exs. It is not an error,
just a warning. It might be a desired behaviour for your case to have such items, because they may be hidden by some conditions
which depend on other variables.
	✓ - variable was provided and it is valid according to schema.
	- - variable wasn't provided and default value is used.

There is a block MISSING SCHEMA DEFINITIONS at the bottom.
It simply prints variable names which were invoked in runtime.exs, but not present in the schema.
Let's make everything functional.
Add DATABASE_NO_SSL to the schema:
  "DATABASE_NO_SSL" => %{
    aliases: ["DB_NO_SSL"],
    cast: Type.Boolean,
    default: "false",
    context: [env: [:prod]],
    groups: [:primary_db]
  }
Set DATABASE_URL in runtime.exs:
config :my_app,
  # ...
  database_url: Confispex.get("DATABASE_URL"),
and run report with valid values:
LOG_LEVEL=info CONTACT_US_EMAILS=myemail1@example.com,myemail2@host DB_URL=postgres://user:pwd@localhost:5432/db_name MIX_ENV=prod mix confispex.report --mode=detailed
[image: state 2]
Verifying Schema Completeness
To ensure all variables accessed in your application are properly documented in the schema,
use mix confispex.check in your CI/CD pipeline:
# Check that all Confispex.get/1 calls reference variables in schema
$ MIX_ENV=prod mix confispex.check
✓ All configuration variables are defined in schema

If you access a variable that's not in your schema, the check will fail:
$ MIX_ENV=prod mix confispex.check
✗ Found variables missing from schema:
  - UNDOCUMENTED_VAR
** (Mix.Error) Configuration check failed: 1 variable(s) not defined in schema

Best Practice: Run this check for all environments in CI, since context filters
may cause different variables to be available in different environments:
MIX_ENV=dev mix confispex.check
MIX_ENV=test mix confispex.check
MIX_ENV=prod mix confispex.check

This prevents developers from forgetting to document new variables when adding them
to config/runtime.exs.


  

    
Confispex 
    



      
A tool which allows defining specs for runtime configuration, cast values according to specified types and inspect them.
Workflow
	Define a schema - Create a module with defvariables/1 describing your configuration
	Initialize - Call init/1 in config/runtime.exs with your schema and context
	Get values - Use get/1 to retrieve typed and validated configuration values
	Inspect - Run mix confispex.report to see all variables and their status

Key Concepts
	Schema - Defines variables with types, defaults, validation rules, and grouping
	Context - Runtime environment info (e.g., %{env: :prod, target: :host})
	Store - Source of raw configuration values (default: System.get_env/0)
	Groups - Logical organization of related variables for reporting
	Invocations - Tracked variable access for comprehensive error reporting

Example
# 1. Define schema
defmodule MyApp.ConfigSchema do
  import Confispex.Schema
  @behaviour Confispex.Schema

  defvariables(%{
    "DATABASE_URL" => %{
      cast: Confispex.Type.URL,
      required: [:database],
      groups: [:database]
    }
  })
end

# 2. Initialize in config/runtime.exs
Confispex.init(%{
  schema: MyApp.ConfigSchema,
  context: %{env: config_env()}
})

# 3. Use values
config :my_app, MyApp.Repo,
  url: Confispex.get("DATABASE_URL")

# 4. Inspect (command line)
# $ mix confispex.report --mode=detailed
See the Getting Started guide for more details.

        context()

      


        Runtime context information used for conditional defaults and requirements.



    


    
      
        store()

      


        A map containing configuration values, typically environment variables.



    





  
    Functions
  


    
      
        all_required_touched?(group_name, server \\ Confispex.Server)

      


        Returns true if all required variables in specified group are present in store.



    


    
      
        any_required_touched?(group_name, server \\ Confispex.Server)

      


        Returns true if any required variable in specified group is present in store.



    


    
      
        get(variable_name, server \\ Confispex.Server)

      


        Get a value from store by specified variable name (key) and cast it according to schema.



    


    
      
        init(params, server \\ Confispex.Server)

      


        Initialize or reinitialize a state in server



    


    
      
        init_once(params, server \\ Confispex.Server)

      


        Initialize a state in server if it hasn't already been initialized.



    


    
      
        report(mode, opts \\ [])

      


        Print report with variables usage to STDOUT.



    


    
      
        update_store(update_fn, server \\ Confispex.Server)

      


        Update the store at runtime by applying a function to the current store.

          @type context() :: %{required(atom()) => atom()}


      


Runtime context information used for conditional defaults and requirements.
Commonly includes :env (:dev, :test, :prod) and :target (:host, :docker).

          @type store() :: map()


      


A map containing configuration values, typically environment variables.
Keys are variable names (usually strings) and values are their string representations.

      all_required_touched?(group_name, server \\ Confispex.Server)



    

  


  

      

          @spec all_required_touched?(group_name :: atom(), GenServer.server()) :: boolean()


      


Returns true if all required variables in specified group are present in store.
Use this when you want to ensure ALL required variables are configured before
enabling a feature. This is stricter than any_required_touched?/1.
Example
# Only configure database if ALL required variables are provided
if Confispex.all_required_touched?(:database) do
  config :my_app, MyApp.Repo,
    url: Confispex.get("DATABASE_URL"),
    pool_size: Confispex.get("DATABASE_POOL_SIZE"),
    ssl: Confispex.get("DATABASE_SSL")
end
Compare with any_required_touched?/1: if even ONE required variable is present,
any_required_touched?/1 returns true. This function requires ALL of them.

  



    

  
    
      
    
    
      any_required_touched?(group_name, server \\ Confispex.Server)



    

  


  

      

          @spec any_required_touched?(group_name :: atom(), GenServer.server()) :: boolean()


      


Returns true if any required variable in specified group is present in store.
Use this to detect if the user is trying to configure a group, even if the
configuration is incomplete. This is useful for conditional configuration of
services that crash on invalid config (better to skip than crash).
Example
# Configure APNS only if user provided at least one APNS variable
if Confispex.any_required_touched?(:apns) do
  config :pigeon, :apns,
    sandbox: %{
      cert: Confispex.get("APNS_CERT"),
      key: Confispex.get("APNS_KEY"),
      mode: :dev
    }
end
Difference from all_required_touched?/1
	any_required_touched?/1 - returns true if at least one required variable is present
(user is trying to configure this group)
	all_required_touched?/1 - returns true only if all required variables are present
(configuration is complete)

Use any_required_touched?/1 to decide "should I configure this at all?" and
all_required_touched?/1 to decide "is configuration complete and valid?"

  



    

  
    
      
    
    
      get(variable_name, server \\ Confispex.Server)



    

  


  

      

          @spec get(Confispex.Schema.variable_name(), GenServer.server()) :: any()


      


Get a value from store by specified variable name (key) and cast it according to schema.
Returns the casted value on success, or nil on any error (variable not found, type
casting failed, etc.). Errors are collected and can be viewed with report/1.
Example
config :my_app, MyApp.Repo, url: Confispex.get("DATABASE_URL")
Behavior
	Variable found and valid: returns the casted value
	Variable found via alias: returns the casted value (e.g., DB_URL when DATABASE_URL not found)
	Variable not found but has default: returns the default value (casted)
	Type casting fails: returns nil and saves error for later reporting
	Variable not in schema: returns nil and shows warning in report

To see all errors, run:
mix confispex.report --mode=detailed
# or
Confispex.report(:detailed)
This design allows your application to start even with configuration errors, so you
can see ALL problems at once in the report, rather than fixing them one-at-a-time.

  



    

  
    
      
    
    
      init(params, server \\ Confispex.Server)



    

  


  

      

          @spec init(
  %{
    :schema => module(),
    :context => context(),
    optional(:store) => store() | (-> store())
  },
  GenServer.server()
) :: :ok


      


Initialize or reinitialize a state in server
Example
Confispex.init(%{
  schema: MyApp.RuntimeConfigSchema,
  context: %{env: config_env(), target: config_target()}
})
By default, Confispex uses System.get_env/0 to setup the store.

  



    

  
    
      
    
    
      init_once(params, server \\ Confispex.Server)



    

  


  

      

          @spec init_once(
  %{
    :schema => module(),
    :context => context(),
    optional(:store) => store() | (-> store())
  },
  GenServer.server()
) :: :ok


      


Initialize a state in server if it hasn't already been initialized.
Use this instead of init/1 when you want to initialize only once, ignoring
subsequent calls. This is useful when configuration file is re-read with
Config.Reader.read!/1 to prevent overwriting the existing state.
Example
# In config/runtime.exs - initializes only on first call
Confispex.init_once(%{
  schema: MyApp.RuntimeConfigSchema,
  context: %{env: config_env(), target: config_target()}
})
By default, Confispex uses System.get_env/0 to setup the store.

  



    

  
    
      
    
    
      report(mode, opts \\ [])



    

  


  

      

          @spec report(
  :detailed | :brief,
  keyword()
) :: :ok


      


Print report with variables usage to STDOUT.
The report shows all variables organized by groups with color-coded status:
	Green groups: all required variables present and valid
	Red groups: required variables missing or invalid
	Blue groups: functional (no required variables or all have defaults)

Modes
	:detailed - shows actual values from the store (may contain sensitive data)
	:brief - hides values, only shows variable status (safe for logs)

Options
	:server (atom/0) - The GenServer to query for report data. Defaults to the internal server.

	:emit_ansi? (boolean/0) - Whether to emit ANSI color codes. Defaults to IO.ANSI.enabled?().


Examples
# Show full report with values
Confispex.report(:detailed)

# Show report without values (safe for CI/logs)
Confispex.report(:brief)

# Force colors on remote shell
Confispex.report(:detailed, emit_ansi?: true)

# Custom server with colors disabled
Confispex.report(:brief, server: MyApp.ConfigServer, emit_ansi?: false)
You can also use the mix task:
mix confispex.report --mode=detailed
mix confispex.report --mode=brief

  



    

  
    
      
    
    
      update_store(update_fn, server \\ Confispex.Server)



    

  


  

      

          @spec update_store((store() -> store()), GenServer.server()) :: :ok


      


Update the store at runtime by applying a function to the current store.
Example
# Add or override specific variables
new_values = %{"FEATURE_X_ENABLED" => "true", "API_KEY" => "new-key"}
Confispex.update_store(&Map.merge(&1, new_values))

# Remove a variable
Confispex.update_store(&Map.delete(&1, "TEMP_CONFIG"))

Defines the behavior and helpers for creating configuration schemas.
A schema describes all environment variables your application uses, their types,
defaults, validation rules, and how they're organized into logical groups.
Basic Usage
defmodule MyApp.RuntimeConfigSchema do
  import Confispex.Schema
  @behaviour Confispex.Schema
  alias Confispex.Type

  defvariables(%{
    "DATABASE_URL" => %{
      cast: Type.URL,
      groups: [:database],
      required: [:database]
    }
  })
end
Variable Specification
Each variable is defined with a map containing the following options:
Required Options
	:cast - type specification for casting the value. Can be:
	A module implementing Confispex.Type behavior (e.g., Type.String)
	A tuple with module and options (e.g., {Type.Integer, scope: :positive})


	:groups - list of group atoms this variable belongs to. Groups are used for:
	Organizing variables in reports
	Checking if a feature is properly configured via all_required_touched?/1
	Conditional configuration based on any_required_touched?/1



Optional Options
	:doc - human-readable description shown in generated .envrc template files

	:default - default value as a string. Used when variable is not present in the store.
Cannot be used with :required or :default_lazy.
"PORT" => %{
  cast: Type.Integer,
  default: "4000",
  groups: [:server]
}

	:default_lazy - function (context -> String.t() | nil) for context-dependent defaults.
Receives runtime context (e.g., %{env: :prod, target: :host}) and returns default value
or nil to skip default. Cannot be used with :default.
"LOG_LEVEL" => %{
  cast: {Type.Enum, values: ["debug", "info", "warning", "error"]},
  default_lazy: fn
    %{env: :prod} -> "warning"
    %{env: :dev} -> "debug"
    _ -> "info"
  end,
  groups: [:logging]
}

	:required - marks variable as required in specific groups. Can be:
	List of group atoms (e.g., [:database, :cache])
	Function (context -> [atom()]) for context-dependent requirements

When all required variables in a group are present and valid, the group is considered
ready for use. Cannot be used with :default.
"DATABASE_URL" => %{
  cast: Type.URL,
  groups: [:database],
  required: [:database]  # Required when using :database group
}

	:context - keyword list specifying when this variable should be included in the schema.
Variables outside their context are filtered out completely.
"DATABASE_URL" => %{
  cast: Type.URL,
  context: [env: [:prod]],  # Only in production
  groups: [:database]
}

	:aliases - list of alternative names for this variable. Confispex will try each name
in order until it finds a value in the store.
"DATABASE_URL" => %{
  aliases: ["DB_URL", "DATABASE_CONNECTION"],
  cast: Type.URL,
  groups: [:database]
}

	:template_value_generator - function (-> String.t()) that generates a value for
.envrc template. Useful for secrets that should be generated once.
"SECRET_KEY_BASE" => %{
  cast: Type.String,
  template_value_generator: fn -> :crypto.strong_rand_bytes(64) |> Base.encode64() end,
  groups: [:security]
}


Complete Example
defmodule MyApp.RuntimeConfigSchema do
  import Confispex.Schema
  @behaviour Confispex.Schema
  alias Confispex.Type

  defvariables(%{
    # Simple required variable
    "DATABASE_URL" => %{
      aliases: ["DB_URL"],
      doc: "PostgreSQL connection URL",
      cast: Type.URL,
      context: [env: [:prod]],
      groups: [:database],
      required: [:database]
    },

    # Variable with default
    "DATABASE_POOL_SIZE" => %{
      aliases: ["POOL_SIZE"],
      cast: {Type.Integer, scope: :positive},
      default: "10",
      context: [env: [:prod]],
      groups: [:database]
    },

    # Context-dependent default
    "LOG_LEVEL" => %{
      cast: {Type.Enum, values: ["debug", "info", "warning", "error"]},
      default_lazy: fn
        %{env: :test} -> "warning"
        %{env: :dev} -> "debug"
        %{env: :prod} -> "info"
      end,
      groups: [:logging]
    },

    # Generated secret
    "SECRET_KEY_BASE" => %{
      cast: Type.String,
      template_value_generator: fn ->
        :crypto.strong_rand_bytes(64) |> Base.encode64()
      end,
      groups: [:security],
      required: [:security]
    }
  })
end

        variable_name()

      


        Name of a configuration variable.



    


    
      
        variable_spec()

      


        Specification for a single configuration variable.



    





  
    Callbacks
  


    
      
        variables_schema()

      


    





  
    Functions
  


    
      
        defvariables(variables)

      


        A helper which performs basic validations of the input schema and then defines variables_schema/0 function.

      variable_name()



    

  


  

      

          @type variable_name() :: term()


      


Name of a configuration variable.
Typically a string representing an environment variable name (e.g., "DATABASE_URL"),
but can be any term.

  



  
    
      
    
    
      variable_spec()



    

  


  

      

          @type variable_spec() :: %{
  :cast => module() | {module(), opts :: keyword()},
  :groups => [atom()],
  optional(:doc) => String.t(),
  optional(:default) => String.t(),
  optional(:default_lazy) => (Confispex.context() -> String.t() | nil),
  optional(:template_value_generator) => (-> String.t()),
  optional(:required) => [atom()] | (Confispex.context() -> [atom()]),
  optional(:context) => [{atom(), atom()}],
  optional(:aliases) => [variable_name()]
}


      


Specification for a single configuration variable.
See the module documentation for detailed explanation of each option.

          @callback variables_schema() :: %{required(variable_name()) => variable_spec()}

      defvariables(variables)


        (macro)


    

  


  

A helper which performs basic validations of the input schema and then defines variables_schema/0 function.

Defines the behavior for type casting in Confispex.
All configuration values start as strings (from environment variables) and need to be
cast to appropriate Elixir types. This module defines the behavior for type casting
and provides several built-in type implementations.
Built-in Types
Confispex provides the following built-in types:
	Confispex.Type.Boolean - casts to boolean (true/false)
	Confispex.Type.Integer - casts to integer with optional scope (:positive)
	Confispex.Type.Float - casts to float
	Confispex.Type.String - validates non-empty strings
	Confispex.Type.Enum - validates value against allowed list
	Confispex.Type.Email - basic email format validation
	Confispex.Type.URL - URL format validation
	Confispex.Type.CSV - parse CSV strings with nested type support
	Confispex.Type.JSON - parse JSON objects with atom/string key conversion
	Confispex.Type.Base64Encoded - decode base64 and cast with nested type
	Confispex.Type.Term - evaluate Elixir terms from strings

Using Types in Schema
Types can be referenced as a module or as a tuple with options:
defvariables(%{
  "PORT" => %{
    cast: Confispex.Type.Integer,  # Simple reference
    groups: [:server]
  },
  "LOG_LEVEL" => %{
    cast: {Confispex.Type.Enum, values: ["debug", "info", "warning", "error"]},  # With options
    groups: [:logging]
  }
})
Creating Custom Types
To create a custom type, implement the Confispex.Type behavior:
defmodule MyApp.PortType do
  @behaviour Confispex.Type

  @impl true
  def cast(value, _opts) when is_binary(value) do
    case Integer.parse(value) do
      {port, ""} when port >= 1 and port <= 65535 ->
        {:ok, port}

      _ ->
        {:error, ["expected a valid port number (1-65535)"]}
    end
  end

  def cast(_value, _opts) do
    {:error, ["expected a string"]}
  end
end
Then use it in your schema:
defvariables(%{
  "PORT" => %{
    cast: MyApp.PortType,
    groups: [:server]
  }
})
Error Details Structure
Error details use a tagged structure for rich error reporting with ANSI colors.
The structure is a list that can contain strings and tagged tuples.
Available Tags
	String - plain error message, rendered as-is
{:error, ["value must be a positive number"]}

	{:highlight, text} - highlighted values (rendered in light cyan, typically used
for showing valid options or values)
{:error, [
  "expected one of: ",
  {:highlight, "prod"},
  ", ",
  {:highlight, "dev"},
  ", ",
  {:highlight, "test"}
]}

	{:validation, details} - validation error section (prefixed with "Validation failed: "
in light red). Use this when the input format is correct but doesn't meet business rules
or constraints.
# Example: Value was successfully parsed as integer, but doesn't meet range requirement
{:error, [
  {:validation, ["port must be between 1 and 65535"]}
]}

	{:parsing, details} - parsing error section (prefixed with "Parsing failed: " in
light red). Use this when the input cannot be converted to the expected type or format.
# Example: Value cannot be parsed as an integer at all
{:error, [
  {:parsing, ["failed to parse integer from ", {:highlight, "abc123"}]}
]}

	{:nested, [details]} - nested type errors (used when a collection type like CSV
needs to report errors from multiple items). Rendered with "Casting nested elements
failed: " header in light red. The details list contains full error tuples from
inner types: {value, type, error_details}.
# Example: CSV type trying to parse a list of emails, where one email is invalid
{:error,
 {"admin@,user", {Confispex.Type.CSV, [of: Confispex.Type.Email]},
  [
    nested: [
      {"admin@", Confispex.Type.Email,
       [validation: ["incomplete email address"]]}
    ]
  ]}}


When to Use :nested
Use :nested when your type needs to report errors from multiple items or add context
about WHERE the error occurred in a complex structure.
Main use case: Collection types - when validating multiple items (like CSV):
# Real example from Confispex.Type.CSV implementation
results =
  Enum.map(items, fn item ->
    Confispex.Type.cast(item, inner_type)
  end)

case Enum.filter(results, &match?({:error, _}, &1)) do
  [] ->
    {:ok, Enum.map(results, &elem(&1, 1))}

  errors ->
    {:error, nested: Enum.map(errors, &elem(&1, 1))}
end

# This allows reporting ALL failed items at once, not just the first one
Another use case: Adding context - when you want to add information about where
the error occurred. Note: This is less common - usually you'd just pass the error through.
# Example: Adding context to an inner type's error (rarely needed)
case Confispex.Type.cast(value, inner_type) do
  {:ok, result} ->
    {:ok, result}

  {:error, {failed_value, type, details}} ->
    # Wrap the error to add context
    {:error, nested: [{failed_value, type, ["in field 'port': " | details]}]}
end
Parsing vs Validation
Understanding when to use :parsing vs :validation:
:parsing - structural/format errors (cannot convert to target type):
	"abc" → Integer (not a number at all)
	"not-an-email" → Email (doesn't contain @)
	"not-json" → JSON (invalid JSON syntax)
	"invalid-base64!@#" → Base64 (invalid base64 characters)

:validation - business rule errors (converted successfully but doesn't meet constraints):
	"99999" → Integer with scope :positive and max 65535 (too large)
	"staging" → Enum with values ["dev", "prod"] (not in allowed list)
	"user@" → Email (has @ but incomplete)
	"-5" → Integer with scope :positive (negative number)

Complex Example
Here's a real-world error from CSV type with nested Email validation:
# When CSV is parsed successfully, but one of the emails inside is invalid
{:error,
 {"John,user1@example.com", {Confispex.Type.CSV, [of: Confispex.Type.Email]},
  [
    nested: [
      {"John", Confispex.Type.Email,
       [parsing: ["expected a string in format ", {:highlight, "username@host"}]]}
    ]
  ]}}
This shows:
	CSV was parsed successfully (no :parsing error at top level)
	Inside the CSV (:nested), one item failed: "John"
	That item failed Email type's parsing (:parsing) - doesn't match email format
	The nested error is a full error tuple: {value, type, details}

Note: If CSV itself fails to parse, you get :parsing instead of :nested:
# When CSV structure is broken (e.g., unclosed quotes)
{:error, [
  {:parsing, ["expected escape character " but reached the end of file"]}
]}
The type returns either :parsing (own error) or :nested (inner type error),
never both at the same time.
ANSI Color Reference
When rendered in terminal reports:
	:highlight tags - light cyan
	:validation and :parsing prefixes - light red
	:nested header - light red

          @type error_details() :: [
  String.t()
  | error_details()
  | {:highlight | :validation | :parsing, error_details()}
  | {:nested, [error_details()]}
]


      


Structured error details returned when type casting fails.
Supports nested structures with formatting hints for terminal output.

          @type type_reference() :: module() | {module(), opts :: Keyword.t()}


      


A type specification for casting configuration values.
Can be either:
	A module implementing Confispex.Type behavior (e.g., Confispex.Type.String)
	A tuple with module and options (e.g., {Confispex.Type.Integer, scope: :positive})

          @callback cast(value :: term(), opts :: Keyword.t()) ::
  {:ok, value :: term()} | :error | {:error, error_details()}

          @spec cast(input :: any(), type_reference()) ::
  {:ok, output :: any()}
  | {:error, {failed_on_value :: any(), type_reference(), error_details()}}


      


Cast input using type in type_reference()
Examples
iex> Confispex.Type.cast("dev", {Confispex.Type.Enum, values: [:prod, :test, :dev]})
{:ok, "dev"}

iex> Confispex.Type.cast("prodd", {Confispex.Type.Enum, values: [:prod, :test, :dev]})
{:error,
 {"prodd", {Confispex.Type.Enum, [values: [:prod, :test, :dev]]},
  [
    validation: [
      "expected one of: ",
      [
        {:highlight, "prod"},
        ", ",
        {:highlight, "test"},
        ", ",
        {:highlight, "dev"}
      ]
    ]
  ]}}

  


        

      


  

    
Confispex.Type.Base64Encoded 
    



      
A type for base64 encoded values.
Decodes base64 encoded string.
Options
	:of - Type reference for casting decoded value. Can be a module or {module, opts} tuple. Defaults to Confispex.Type.String. The default value is Confispex.Type.String.

Examples
iex> Confispex.Type.cast("aGVsbG8=", Confispex.Type.Base64Encoded)
{:ok, "hello"}

iex> Confispex.Type.cast("//8=", Confispex.Type.Base64Encoded)
{:error, {<<255, 255>>, Confispex.Type.String, [validation: "not a valid string"]}}

iex> Confispex.Type.cast("//8=", {Confispex.Type.Base64Encoded, of: Confispex.Type.Term})
{:ok, <<0xFFFF::16>>}

      




  

    
Confispex.Type.Boolean 
    



      
A boolean type.
Casts "enabled", "true", "1", "yes", 1, true to true,
and "disabled", "false", "0", "no", 0, false to false.
Options
This type has no options.
Examples
iex> Confispex.Type.cast("yes", Confispex.Type.Boolean)
{:ok, true}

iex> Confispex.Type.cast("true", Confispex.Type.Boolean)
{:ok, true}

iex> Confispex.Type.cast("0", Confispex.Type.Boolean)
{:ok, false}

iex> Confispex.Type.cast("disabled", Confispex.Type.Boolean)
{:ok, false}

iex> Confispex.Type.cast("DISABLED", Confispex.Type.Boolean)
{
  :error,
  {"DISABLED", Confispex.Type.Boolean,
   [
     validation: [
       "expected one of: ",
       [
         {:highlight, "enabled"},
         ", ",
         {:highlight, "true"},
         ", ",
         {:highlight, "1"},
         ", ",
         {:highlight, "yes"},
         ", ",
         {:highlight, "disabled"},
         ", ",
         {:highlight, "false"},
         ", ",
         {:highlight, "0"},
         ", ",
         {:highlight, "no"}
       ]
     ]
   ]}
}

      




  

    
Confispex.Type.CSV 
    



      
A CSV type.
Casts a CSV string to a list with values which are cast according to :of option.
Options
	:of - Type reference for casting CSV values. Can be a module or {module, opts} tuple. Defaults to Confispex.Type.String. The default value is Confispex.Type.String.

Examples
iex> Confispex.Type.cast("John,user1@example.com", Confispex.Type.CSV)
{:ok, ["John", "user1@example.com"]}

iex> Confispex.Type.cast("John,user1@example.com", {Confispex.Type.CSV, of: Confispex.Type.Email})
{:error,
 {"John,user1@example.com", {Confispex.Type.CSV, [of: Confispex.Type.Email]},
  [
    nested: [
      {"John", Confispex.Type.Email,
       [parsing: ["expected a string in format ", {:highlight, "username@host"}]]}
    ]
  ]}}

iex> Confispex.Type.cast(~s|John,"user1@example.com|, Confispex.Type.CSV)
{:error,
 {~s|John,"user1@example.com|, Confispex.Type.CSV,
  [parsing: ~s|expected escape character " but reached the end of file|]}}

      




  

    
Confispex.Type.Email 
    



      
An email type.
Returns input value if it is in username@host format.
Options
This type has no options.
Examples
iex> Confispex.Type.cast("user@example.com", Confispex.Type.Email)
{:ok, "user@example.com"}

iex> Confispex.Type.cast("user[at]example.com", Confispex.Type.Email)
{:error,
 {"user[at]example.com", Confispex.Type.Email,
  [parsing: ["expected a string in format ", {:highlight, "username@host"}]]}}

      




  

    
Confispex.Type.Enum 
    



      
An enum type.
Returns input value if it is present in a list specified in :values option.
Options
	:values (list of term/0) - Required. A list of allowed values. Can contain any values which implement String.Chars protocol.

Examples
iex> Confispex.Type.cast("low", {Confispex.Type.Enum, values: ["low", "high"]})
{:ok, "low"}

iex> Confispex.Type.cast("LOW", {Confispex.Type.Enum, values: ["low", "high"]})
{:error,
 {"LOW", {Confispex.Type.Enum, [values: ["low", "high"]]},
  [validation: ["expected one of: ", [{:highlight, "low"}, ", ", {:highlight, "high"}]]]}}

      




  

    
Confispex.Type.Float 
    



      
A float type.
Casts input string to Float.
Options
This type has no options.
Examples
iex> Confispex.Type.cast("3.14", Confispex.Type.Float)
{:ok, 3.14}

iex> Confispex.Type.cast("314", Confispex.Type.Float)
{:ok, 314.0}

iex> Confispex.Type.cast("1.0e2", Confispex.Type.Float)
{:ok, 100.0}

iex> Confispex.Type.cast("1.0ee2", Confispex.Type.Float)
{:error,
 {"1.0ee2", Confispex.Type.Float, [parsing: ["unexpected substring ", {:highlight, ~s|"ee2"|}]]}}

      




  

    
Confispex.Type.Integer 
    



      
An integer type.
Casts input string to Integer.
Options
	:scope - Requires parsed integer value to be positive (> 0).

Examples
iex> Confispex.Type.cast("-42", Confispex.Type.Integer)
{:ok, -42}

iex> Confispex.Type.cast("-42", {Confispex.Type.Integer, scope: :positive})
{:error, {"-42", {Confispex.Type.Integer, [scope: :positive]}, [validation: "expected a positive integer"]}}

iex> Confispex.Type.cast("42 monkeys", Confispex.Type.Integer)
{:error,
 {"42 monkeys", Confispex.Type.Integer,
  [parsing: ["unexpected substring ", {:highlight, ~s|" monkeys"|}]]}}

      




  

    
Confispex.Type.JSON 
    



      
A JSON type.
Casts JSON string to Elixir terms using built-in JSON module.
Options
	:keys - How to handle map keys. Options:	:strings - keys remain as strings (default)
	:atoms - converts keys to existing atoms only (safe)
	:atoms! - converts keys to atoms, creating new atoms if needed (use with caution) The default value is :strings.



Examples
iex> Confispex.Type.cast(~s|[{"email":"john@example.com","level":1}]|, Confispex.Type.JSON)
{:ok, [%{"email" => "john@example.com", "level" => 1}]}

iex> Confispex.Type.cast(~s|[{"email":"john@example.com","level":1}]|, {Confispex.Type.JSON, keys: :atoms})
{:ok, [%{email: "john@example.com", level: 1}]}

iex> Confispex.Type.cast("", Confispex.Type.JSON)
{:error, {"", Confispex.Type.JSON, [parsing: "unexpected end of input at position 0"]}}

      




  

    
Confispex.Type.String 
    



      
A string type.
Returns input string if it is not empty.
Options
This type has no options.
Examples
iex> Confispex.Type.cast("value", Confispex.Type.String)
{:ok, "value"}

iex> Confispex.Type.cast("", Confispex.Type.String)
{:error, {"", Confispex.Type.String, [validation: "blank string"]}}

iex> Confispex.Type.cast("value" <> <<0xFFFF::16>>, Confispex.Type.String)
{:error, {"value" <> <<0xFFFF::16>>, Confispex.Type.String, [validation: "not a valid string"]}}

      




  

    
Confispex.Type.Term 
    



      
Represents any term.
Returns input.
Options
This type has no options.
Examples
iex> Confispex.Type.cast("value", Confispex.Type.Term)
{:ok, "value"}

iex> Confispex.Type.cast("", Confispex.Type.Term)
{:ok, ""}

      




  

    
Confispex.Type.URL 
    



      
An URL type.
Returns input string if it is a valid URL.
Options
This type has no options.
Examples
iex> Confispex.Type.cast("postgres://user@pass:host", Confispex.Type.URL)
{:ok, "postgres://user@pass:host"}

iex> Confispex.Type.cast("localhost", Confispex.Type.URL)
{:error, {"localhost", Confispex.Type.URL, [validation: "missing a scheme (e.g. https)"]}}

      




  

    
mix confispex.check 
    



      
Checks that all configuration variables accessed via Confispex.get/1 are
defined in your schema. Exits with error code 1 if any variables are missing
from the schema.
This task is designed for CI/CD pipelines to prevent runtime issues caused by
accessing undocumented configuration variables.
Usage in CI/CD
Add this to your CI pipeline after running tests.
Important: Always run this check for each environment you use (dev, test, prod)
since different environments may have different variables due to context filters.
# In your CI config (e.g., GitHub Actions, GitLab CI)
$ MIX_ENV=dev mix confispex.check
$ MIX_ENV=test mix confispex.check
$ MIX_ENV=prod mix confispex.check
The task will:
	Exit with code 0 if all variables are defined in schema
	Exit with code 1 if any variables are missing from schema
	Print the list of missing variables to stderr

Examples
# Check for missing schema definitions
$ MIX_ENV=prod mix confispex.check

# Example output on success:
✓ All configuration variables are defined in schema

# Example output on failure:
✗ Found variables missing from schema:
  - UNDOCUMENTED_VAR_1
  - UNDOCUMENTED_VAR_2
When to Use
Run this task in CI/CD after:
	Your application has started and configuration has been loaded
	All Confispex.get/1 calls have been made (usually during app startup)

This ensures developers don't forget to document new variables when adding
them to config/runtime.exs.

mix confispex.gen.doc.md 
    



      
Generates a Markdown documentation file listing all configuration variables
from your schema in a human-readable table format.
Perfect for:
	Project README files
	Deployment documentation
	Onboarding new team members
	CI/CD pipeline documentation

The generated document includes tables organized by groups showing:
	Variable names
	Required/optional status
	Default values
	Descriptions from :doc fields

Options
	--schema - (required) the module name of your schema (e.g., MyApp.RuntimeConfigSchema)
	--output - (required) path to output file (e.g., RUNTIME_CONFIG.md)
	--env - (optional) environment context (e.g., prod, dev, test)
	--target - (optional) target context (e.g., host, docker)

Examples
# Generate docs for production environment
$ mix confispex.gen.doc.md \
    --schema=MyApp.RuntimeConfigSchema \
    --output=docs/PRODUCTION_CONFIG.md \
    --env=prod \
    --target=host

# Generate docs for development
$ mix confispex.gen.doc.md \
    --schema=MyApp.RuntimeConfigSchema \
    --output=README_CONFIG.md \
    --env=dev
Example Output
# Variables (env=prod target=host)

## GROUP :database

| Name          | Required | Default   | Description            |
| ------------- | -------- | --------- | ---------------------- |
| DATABASE_URL  | required |           | PostgreSQL connection  |
| DATABASE_POOL |          | 10        | Connection pool size   |

## GROUP :cache

| Name       | Required | Default     | Description       |
| ---------- | -------- | ----------- | ----------------- |
| REDIS_URL  | required |             | Redis connection  |
Context-Aware Documentation
The generated documentation respects context filters in your schema. Variables
marked with context: [env: [:prod]] will only appear in documentation generated
with --env=prod.

mix confispex.gen.template.envrc 
    



      
Generates a .envrc template file from your configuration schema for use with direnv.
The generated file contains all variables from your schema organized by groups,
with helpful comments and smart defaults:
	Required variables - uncommented, ready to fill in
	Variables with defaults - commented out with their default values
	Variables with generators - uncommented with generated values (e.g., secrets)
	Optional variables - commented out

Options
	--schema - (required) the module name of your schema (e.g., MyApp.RuntimeConfigSchema)
	--output - (required) path to output file (e.g., .envrc or .envrc.example)

Examples
# Generate .envrc for development
$ mix confispex.gen.template.envrc --schema=MyApp.RuntimeConfigSchema --output=.envrc

# Generate template for new developers
$ mix confispex.gen.template.envrc --schema=MyApp.RuntimeConfigSchema --output=.envrc.example
Usage with direnv
After generating the file:
	Edit .envrc and fill in required values
	Run direnv allow to load the environment
	Variables will be automatically loaded when you enter the project directory

Safety
The task will prompt for confirmation before overwriting existing files.

Callback implementation for Mix.Task.run/1.

  


        

      


  

    
mix confispex.report 
    



      
Prints a color-coded report of all configuration variables defined in your schema.
The report shows variables organized by groups with their status:
	Green groups: all required variables present and valid
	Red groups: required variables missing or invalid
	Blue groups: functional (no required variables or all have defaults)

Options
	--mode - report mode (default: brief)	brief - shows only variable status without values (safe for CI/logs)
	detailed - shows actual values from the store (may contain sensitive data)



Examples
# Brief report (default, safe for logs)
$ mix confispex.report

# Brief report (explicit)
$ mix confispex.report --mode=brief

# Detailed report with values
$ mix confispex.report --mode=detailed
Usage in Development
Run this task after initializing your app to verify all required environment
variables are properly configured. Use --mode=brief in CI/production logs
to avoid exposing sensitive values.
See Confispex.report/2 for interactive usage (IEx, remote shell).

